Sample solutions for Problems 7, 8 and 10 in Problem Set 12 with `Macaulay2`

Chris Peterson and Hirotachi Abo

Problem 7, 8 and 10 (Set 12). Let C be an irreducible plane curve in $\mathbb{P}^2$ over an algebraically closed field k, that is, the zero locus of an irreducible polynomial $F \in k[x, y, z]$. Recall that the tangent line $T_p(C)$ to C at a point $p \in C$ is defined by the following single polynomial equation:

$$xF_x(p) + yF_y(p) + zF_z(p) = 0.$$

This allows us to define the map φ from C to the dual projective space $(\mathbb{P}^2)^*$ by

$$\varphi(p) = \text{the dual of } T_p(C) = [F_x(p) : F_y(p) : F_z(p)].$$

Let C^* be the projective closure of $\varphi(C)$ in $(\mathbb{P}^2)^*$. This C^* is called the *dual curve* of C.

Question. How can we compute the homogeneous polynomial whose zero locus is the dual curve C^*?

Answer. Let $I = (X-F_x, Y-F_y, Z-F_z)+(F)$ be an ideal in $K[x, y, z, X, Y, Z]$. The generators of this ideal are not homogeneous, unless $\deg(F) = 2$. So we need to introduce *weights* on the variables x, y, z, X, Y, Z. Suppose that $\deg(F) = d$. Then we say that the variables x, y and z have weight 1 and X, Y and Z have weight $d - 1$. Compute a gröbner G basis for I in terms of the lexicographic order. The intersection of G and the ring $k[X, Y, Z]$ is a basis of $I' = I \cap k[X, Y, Z]$. Note that I' has a weighted homogeneous basis. Since the variables X, Y and Z have the same degree, I' is homogeneous in the usual sense. This I' is the defining ideal of the dual curve C^*.

Since we proceed the same operation to solve the problems 7 (b), 8 (b) and 10 in Problem Set 12, we make a function to compute the dual curve for a given plane curve. The name of this function is `idealOfDualCurve`. This function takes as an input the ideal of a given plane curve and returns the ideal of its dual curve. Here is the code of the function `idealOfDualCurve`:

```
i1 : KK=QQ;

i2 : idealOfDualCurve=(idl)->(
          -- Compute the jacobian matrix of a given ideal idl.
          jj:=ideal jacobian idl;
```

```
          deg:=(degree jj_0)#0;
          p2':=KK[X,Y,Z];
          -- p2' is the projective space of the dual space.
          -- Define the ring k[x,y,z,X,Y,Z]
          -- and specify the degrees of the generators
          -- of this ring with Degrees.
          p2xp2:=KK[x,y,z,X,Y,Z,Degrees=>{3:1,3:deg},
          MonomialOrder=>Lex];
          -- Define the ideal I.
          gr:=ideal(substitute(vars p2',p2xp2)
          -gens substitute(jj,p2xp2))+substitute(idl,p2xp2);
          -- Eliminate the first three variables.
          sgr:=ideal selectInSubring(3,gens gb gr);
          -- Substitute sgr into p2'
          didl:=substitute(sgr,p2')
          )

o2 = idealOfDualCurve

o2 : Function
```

Define the homogeneous coordinate ring `ringP2` of $\mathbb{P}^2$:

```
i3 : ringP2=KK[x,y,z]

o3 = ringP2

o3 : PolynomialRing
```

Define the ideals $I = (x^2 + 3xy + y^2)$ and $J = (x^3 + 3xyz + y^2z + z^3)$:

```
i4 : I=ideal(x^2+3*x*z+y^2)

             2    2
o4 = ideal(x  + y  + 3x*z)

o4 : Ideal of ringP2

i5 : J=ideal(x^3+3*x*y*z+y^2*z+z^3)

             3             2     3
o5 = ideal(x  + 3x*y*z + y z + z )
```

```
o5 : Ideal of ringP2
```

Compute the homogeneous ideal of the dual curve C^* of $V(I)$:

```
i6 : I'=idealOfDualCurve(I)

             3  2          1  2
o6 = ideal(-*Y  + X*Z - -*Z )
             4             3

o6 : Ideal of QQ [X, Y, Z]
```

The generator of I' is a quadratic polynomial, as we expected (see Problem 2 in Problem Set 15). Is this curve singular (Problem 7 (c))? To check this, we need the following criterion: Let K be the ideal generated by the entries of the jacobian matrix of I'. Then C^* is singular if and only if $V(K+I') = \emptyset$. The latter condition is equivalent to the condition that there is a positive integer N such that $(X, Y, Z)^N \subseteq K+I'$ (weak Nullstellensatz). This can be checked as follows: Compute the radical of $K+I'$ and see whether $\mathrm{rad}(K+I')$ contains the ideal (X, Y, Z). If $\mathrm{rad}(K+I')$ contains (X, Y, Z), then $(X, Y, Z)^N \subseteq K+I'$ for some N (why?):

```
i7 : singI'=ideal(jacobian I')+I'

               3        2    3  2          1  2
o7 = ideal (Z, -*Y, X - -*Z, -*Y  + X*Z - -*Z )
               2        3    4             3

o7 : Ideal of QQ [X, Y, Z]

i8 : radical singI'

o8 = ideal (Z, Y, X)

o8 : Ideal of QQ [X, Y, Z]
```

In this case, the radical of $K + I'$ is the ideal (X, Y, Z). So we can conclude that C^* is nonsingular.

Next we compute the ideal of the dual curve D^* of $V(J)$:

```
i9 : J'=idealOfDualCurve(J)
```

```
             6     5       4 2       3 3    27   6      3 2    81   2 3    81    4
o9 = ideal(X  - 9X Y + 27X Y  - 27X Y  + --*Y  - 9X Y Z + --*X Y Z - --*X*Y Z -
                                         4               2          2
```

```
o9 : Ideal of QQ [X, Y, Z]
```

Let us recall the following fact:

Proposition. A nonsingular plane cubic curve has nine flex points.

The curve D is a nonsingular cubic curve (we can check this by using the same method as in Problem 7 (c)). The above proposition implies that D has nine flex points. These points correspond to singular points of D^*. There are no bitangent lines to D, because any tangent line to D meets D in $3 = 3 \cdot 1$ points (Bezout's theorem). So the singular locus of D^* is expected to have degree 9(=the number of flex points). Compute the ideal of the singular locus by using the jacobian matrix of J:

```
i10 : singJ'=ideal(jacobian J')+J'

               5      4        3 2       2 3      2 2          3    81   4      3 2
o10 = ideal (6X  - 45X Y + 108X Y  - 81X Y  - 27X Y Z + 81X*Y Z - --*Y Z - 9X Z
                                                                 2

o10 : Ideal of QQ [X, Y, Z]
```

Take the radical of the ideal and compute the degree of this radical with degree:

```
i11 : radSingJ'=radical singJ'

                 2   3  3   1  2    3          1     2   1     2   4   2  3     2
o11 = ideal (X*Y  - -*Y  - -*X Z + -*X*Y*Z - -*X*Z  + -*Y*Z , Y  - -*X Z + X Y*Z
                    2      2       2         3        2             9

o11 : Ideal of QQ [X, Y, Z]

i12 : degree radSingJ'

o12 = 9
```

Let C be a plane curve. A natural question for us is, "what is the dual curve C^{**} of C^*? Is C^{**} maybe the original curve C (Problem 10)? Let C be the smooth conic given in Problem 7. We check $C^{**} = C$. First of all, we have to change the variables of the ring of C^*:

```
i13 : ringP2xP2=KK[x,y,z,X,Y,Z]

o13 = ringP2xP2

o13 : PolynomialRing

i14 : K=substitute(substitute(I',ringP2xP2),{X=>x,Y=>y,Z=>z})

            3  2          1  2
o14 = ideal(-*y  + x*z - -*z )
            4             3

o14 : Ideal of ringP2xP2

i15 : K'=substitute(K,ringP2)

            3  2          1  2
o15 = ideal(-*y  + x*z - -*z )
            4             3

o15 : Ideal of ringP2
```

Compute the dual curve with `idealOfDualCurve`:

```
i16 : idealOfDualCurve(K')

             2    2
o16 = ideal(X  + Y  + 3X*Z)

o16 : Ideal of QQ [X, Y, Z]
```

The ideal of C^{**} equals the ideal of the original curve C.